

Chapter 8


Evaluating the Instruction Set
Architecture of H1: Part 2

An argument in call by value can be `&y`. For example,

```
int y = 1;
void fr(int *p) {
    *p = *p + 5;    // accesses y
}
void main() {
    fr(&y);
}
```

Parameter p points to y on entry into fr . y can be accessed by dereferencing p .

FIGURE 8.2



The special case of passing the address of an argument is referred to as *call by address* or *call by reference*.

The program on the next slide contains both call by value and call by reference. fv uses call by value; fr uses call by reference; fvr uses both call by value and call by reference.

FIGURE 8.1

```
1 #include <iostream>
2 using namespace std;
3
4 int y = 1;
5 void fv(int x)
6 {
7     x = x + 5;
8 }
9 void fr(int *p)
10 {
11     *p = *p + 5;    // dereference pointer to access y
12 }
13 void fvr(int x, int *p)
14 {
15     *p = x;
16 }
17 void main()
18 {
19     fv(y);          // call by value-pass value
20     cout << "y = " << y << endl;
21     fr(&y);          // call by reference-pass address
22     cout << "y = " << y << endl;
23     fvr(20, &y);     // call by value and reference
24     cout << "y = " << y << endl;
25 }
```

Let's examine the assembly language needed for the **fv** and **fr** functions. To call **fv** from **main**, we need

```
ld y      ; get value of y
push      ; create and initialize parameter x
call fv
dloc 1    ; remove x
```

To call **fr**, we need

```
ldc y     ; get the address of y
push      ; create and initialize parameter p
call fr
dloc 1    ; remove p
```

On entry into fv, the relative address of x is 1.


x contains the *value* of y.

On entry into fr, the relative address of p is 1.

p *points to* y.

See the next slide.

FIGURE 8.3 a) On entry into fv



b) On entry into fr




FIGURE 8.5

```

1      ; x = x + 5;
2  fv:      ldc    5          ; get 5
3          addr 1          ; add x
4          str    1          ; store in x
5
6          ret
7  ;=====
8          ; *p = *p + 5;
9  fr:      ldr    1          ; get p
10         ldi          ; get *p
11         add    @5          ; add 5
12         push          ; prepare for sti
13         ldr    2          ; get p
14         sti          ; store into *p
15
16         ret
17 ;=====
18         ; *p = x;
19  fvr:     ldr    1          ; get x
20         push          ; prepare for sti
21         ldr    3          ; get p
22         sti          ; store into *p
23
24         ret
25 ;=====
26 main:     ld     y          ; fv(y);
27         push
28         call  fv
29         dloc 1
30
31         ldc    @m0          ; cout << "y = " << y << endl;
32         sout
33         ld     y
34         dout
35         ldc    '\n'
36         aout
37
38         ldc    y          ; fr(&y);
39         push
40         call  fr
41         dloc 1
42
43         ldc    @m1          ; cout << "y = " << y << endl;

```


(continued)

FIGURE 8.5
(continued)

```
44      sout
45      ld    y
46      dout
47      ldc   '\n'
48      aout
49
50      ldc   y                ; fvr(20, &y);
51      push
52      ldc   20
53      push
54      call  fvr
55      dloc  2
56
57      ldc   @m2              ; cout << "y = " << endl;
58      sout
59      ld    y
60      dout
61      ldc   '\n'
62      aout
63
64      halt
65 @5:    dw    5
66 y:     dw    1
67 @m0:   dw    "y = "
68 @m1:   dw    "y = "
69 @m2:   dw    "y = "
70      end    main
```

Relative address of p changes from 1 to 2 during the execution of fr because of the push.

FIGURE 8.4



Reference parameters

- Receives *address* of corresponding argument.
- Preceded by ‘&’ in the parameter list.
- Whenever a reference parameter is used in the called function, the address it contains is dereferenced.

The address of y is passed because z is a reference parameter.

```
void fr (int &z)
{
    z = z + 5;
}
```

```
void main()
{
    fr(y);           // address of y is passed.
}
```

z is dereferenced wherever it is used because z is a reference parameter.

```
void fr(int &z)
{
    z = z + 5;
}
```

FIGURE 8.6

```
1 #include <iostream>
2 using namespace std;
3
4 int y = 1;
5 void fv(int x)           // x is a value parameter
6 {
7     x = x + 5;
8 }
9 void fr(int &z)           // z is a reference parameter
10 {
11     z = z + 5;
12 }
13 void fvr(int x, int &z)  // x is value, z is ref parameter
14 {
15     z = x;
16 }
17 void main()
18 {
19     fv(y);               // call by value
20     cout << "y = " << y << endl; // y is still 1
21     fr(y);               // call by reference
22     cout << "y = " << y << endl; // y is now 6
23     fvr(20, y);
24     cout << "y = " << y << endl; // y is now 20
25 }
```

Conceptual view of reference parameter mechanism

- The argument itself is passed, replacing the corresponding parameter wherever it appears in the called function.
- This is not what really happens, but it is easy to understand.

Line 7 becomes $x = x + 5$; on the 1st call of fr.
It becomes $y = y + 5$; on the second call.

FIGURE 8.7

```
1 #include <iostream>
2 using namespace std;
3
4 int x = 5, y = 6;
5 void fr(int &z)  // z is a reference parameter
6 {
7     z = z + 5;
8 }
9 void main()
10 {
11     fr(x);           // x is the argument
12     fr(y);           // y is the argument
13 }
```

An exception to the rule that whenever a reference parameter is used, the address it contains is dereferenced is illustrated by the program in the next slide.

y is not dereferenced on line 12 because it is also the argument corresponding to the reference parameter z on line 5.

FIGURE 8.8

```
1 #include <iostream>
2 using namespace std;
3
4 int x = 100;
5 void fr(int &z)
6 {
7     z = z + 5;        // adds 5 to x
8 }
9 void ft(int &y)
10 {
11     y = y + 1;        // adds 1 to x
12     fr(y);            // reference parameter now is argument
13 }
14 void main()
15 {
16     ft(x);
17     cout << "x = " << x << endl;
18 }
```

Call by value vs Call by reference

- Call by value is a “one-way street”.
- Call by reference is a “two-way street”.
- Call by value duplicates the argument. It is time and space *inefficient* for large arguments.
- Call by reference passes only the address of the argument.
- Call by reference requires time-consuming dereferencing operations.

How does call by reference handle the following calls?

fr2(x);

fr2(200);

fr2(x + y);

For the 2nd and 3rd calls, the value of the argument is placed on the stack, and then the address of this value is passed.

FIGURE 8.9

```
1 #include <iostream>
2 using namespace std;
3
4 int x = 1, y = 2;
5 void fv2(int z)      // uses call by value
6 {
7     cout << "z = " << z << endl;
8     z = 5;
9 }
10 void fr2(int &z)     // uses call by reference
11 {
12     cout << "z = " << z << endl;
13     z = 5;
14 }
15 void main()
16 {
17     fv2(x);          // z = 1   is displayed
18     fv2(100);        // z = 100 is displayed
19     fv2(x + y);      // z = 3   is displayed
20     fr2(x);          // z = 1   is displayed, x is assigned 5
21     fr2(100);        // z = 100 is displayed
22     fr2(x + y);      // z = 7   is displayed
23 }
```

FIGURE 8.10

```
1 fv2:      ldc  @m0      ; cout << "z = " << z << endl;
2           sout
3           ldr  1
4           dout
5           ldc  '\n'
6           aout
7
8           ldc  5          ; z = 5;
9           str  1
10
11          ret
12 ;=====
13          ; cout << "z = " << z << endl;
14 fr2:      ldc  @m1      ; get address of @m1
15          sout          ; display "z ="
16          ldr  1          ; get z
17          ldi          ; dereference z
18          dout          ; display it
19          ldc  '\n'      ; newline
20          aout
21
```

(continued)

FIGURE 8.10 (continued)

```
22         ldc    5          ; z = 5;
23         push
24         ldr    2
25         sti
26
27         ret
28 ;=====
29 main:    ld     x          ; fv2(x);
30         push
31         call   fv2
32         dloc   1
33
34         ldc    100        ; fv2(100);
35         push
36         call   fv2
37         dloc   1
38
39         ld     x          ; fv2(x+y);
40         add    y
41         push
42         call   fv2
43         dloc   1
44
45         ldc    x          ; fr2(x);
46         push
47         call   fr2
48         dloc   1
49
50         ; fr2(100);
51         ldc    100        ; get 100
52         push    ; create and init implicit var on stack
53         swap    ; get sp
54         st      @spsave    ; save it
55         swap    ; restore sp
56         ld      @spsave    ; get address of implicit var on tos
57         push    ; pass this address to fr2
58         call   fr2
59         dloc   2          ; deallocate parameter and implicit variable
60
61         ; fr2(x+y);
62         ld     x          ; get x
63         add    y          ; add y
```

(continued)

FIGURE 8.10 (continued)

```
64          push          ; create and init implicit var on stack
65          swap          ; get sp
66          st    @spsave  ; save it
67          swap          ; restore sp
68          ld    @spsave  ; get address of implicit var on tos
69          push          ; pass this address to fr2
70          call fr2
71          dloc 2         ; deallocate parameter and implicit variable
72
73          halt
74 x:        dw    1
75 y:        dw    2
76 @m0:      dw    "z = "
77 @m1:      dw    "z = "
78 @spsave:  dw    0
79          end    main
```

Function overloading

The use of the same name for more than one function. The functions with a common name must have parameter lists that differ in type, order, and/or number.

FIGURE 8.11

```
1 #include <iostream>
2 using namespace std;
3
4 int x = 1;
5 void fol()                // fol with no parameters
6 {
7     x = x + 1;
8 }
9 void fol(int n)           // fol with one int parameter
10 {
11     x = x + n;
12 }
13 void fol(int n, int m)    // fol with two int parameters
14 {
15     x = x + n + m;
16 }
17 void main()
18 {
19     fol(10);               // calls 2nd fol function
20     cout << x << endl;
21     fol();                 // calls 1st fol function
22     cout << x << endl;
23     fol(2, 3);             // calls 3rd fol function
24     cout << x << endl;
25 }
```

Implementing function overloading

- Function overloading is implemented by means of *name mangling*.
- With name mangling, each function name at the assembly level has a mangled name. The mangled name includes an encoding of the parameter list.
- Thus, functions with the same name but different parameter lists will have distinct names at the assembly level.

The assembly-level names for

`void fol() {...}`

`void fol(int n) {...}`

`void fol(int n, int m) {...}`

`void fol(int *p) {...}`

`void fol(int &z) {...}`

`void fol(float q) {...}`

are

`@fol$v`

`@fol$i`

`@fol$ii`

`@fol$pi`

`@fol$ri`

`@fol$f`

FIGURE 8.12

Type

Encoding

void	v
boolean	b
signed char	zc
unsigned char	uc
char	compiler dependent: zc or uc depending on how compiler treats char (see Section 1.9)
int	i
signed int	i
unsigned int	ui
short int	s
signed short	s
signed short int	s
unsigned short	us
unsigned short int	us
long	l
long int	l
signed long int	l
unsigned long	ul
unsigned long int	ul
float	f
double	d
long double	g
pointer	prefix p
reference	prefix r

FIGURE 8.13

```
1 @fol$v:    ldc    1                ; x = x + 1;
2           add    x
3           st     x
4
5           ret
6 ;=====
7 @fol$i:    ld     x                ; x = x + n;
8           addr   1
9           st     x
10
11          ret
12 ;=====
13 @fol$ii:   ld     x                ; x = x + n + m;
14          addr   1
15          addr   2
16          st     x
17
18          ret
19 ;=====
20 main:      ldc    10                ; fol(10);
21          push
22          call   @fol$i
23          dloc   1
24
25          ld     x                ; cout << x << endl;
26          dout
27          ldc    '\n'
28          aout
```

(continued)

FIGURE 8.13

(continued)

```
29
30          call @fol$v          ; fol();
31
32          ld    x              ; cout << x << endl;
33          dout
34          ldc   '\n'
35          aout
36
37          ldc   3              ; fol(2, 3);
38          push
39          ldc   2
40          push
41          call @fol$ii
42          dloc  2
43
44          ld    x              ; cout << x << endl;
45          dout
46          ldc   '\n'
47          aout
48
49          halt
50 x:      dw    1
51          end   main
```


C++ struct

- A user-defined type consisting of fields, each of which with its own type.
- The dot operator ('.') is used to access a field of a struct variable, given its name.
- The pointer operator ('->') is used to access a field of a struct variable, given its address.
- Successive fields are mapped to successive locations.

FIGURE 8.14

```
1 struct Coordinates {  
2     int x;  
3     int y;  
4 };  
5 Coordinates gs;           // global struct  
6 void tests(Coordinates *ps)  
7 {  
8     Coordinates ls;       // local struct  
9     int li = 5;           // local int  
10    ls.y = 4;  
11    ps -> y = li;  
12 }  
13 void main()  
14 {  
15     tests(&gs);  
16     gs.y = 3;  
17 }
```

Mangled name for the tests
function in the preceding slide is

@tests\$p11Coordinates

```
struct iii {
```

```
...
```

```
};
```

```
void f(iii x) {
```

```
...
```

```
}
```

What's the mangled name for f?

@f\$iii ← 3 int's or struct iii?

or

@f\$3iii ← not ambiguous

FIGURE 8.15

```
1 @tests$pl1Coordinates:
2         alloc 2           ; Coordinates 1s
3
4         ldc 5             ; int li = 5;
5         push
6
7                               ; ls.y = 4;
8         ldc 4             ; get ls.y
9         str 2             ; store into ls.y
10
11                               ; ps -> y = li;
12         ldr 0             ; get li
13         push             ; push it
14         ldc 1             ; get offset of y field
15         addr 5            ; add ps
16         sti              ; assign li to ps -> y
17
18         dloc 3            ; deallocate locals
19         ret
20 ;=====
21 main:    ldc  gs          ; tests(&gs);
22         push
23         call @tests$pl1Coordinates
24         dloc 1
25
26         ldc 3            ; gs.y = 3
27         st  gs + 1
28
29         halt
30 gs:      dw  2 dup 0
31         end  main
```

4

A struct can be returned with a return statement in C++.

Requires that the address of a return area be passed to the called function.

See the program on the next slide.

FIGURE 8.16

```
1 struct S {  
2     int x;  
3     int y;  
4 };  
5 S t;  
6 S ret_struct() {  
7     S s;  
8     s.x = 1;  
9     s.y = 2;  
10    return s;    // return a struct  
11 }  
12 void main()  
13 {  
15     t = ret_struct();  
16 }
```

FIGURE 8.17

```

1  @ret_struct$V:
2      aloc 2      ; S s
3
4      ldc 1      ; s.x = 1;
5      str 0
6
7      ldc 2      ; s.y = 2;
8      str 1
9
10     ; return s;
11     ldr 1      ; get s.y
12     push      ; push s.y
13     ldr 1      ; get s.x
14     push      ; push s.x
15     ldr 5      ; get address of return area
16     sti      ; pop value of s.x into return area
17     add @1     ; get address of next word in return area
18     sti      ; pop value of s.y into return area
19     dloc 2     ; deallocate s
20     ret
21 ;=====
22 main:
23
24     ; t = ret_struct();
25     ldc t      ; get address of t
26     push      ; create implicit parameter
27     call @ret_struct$V
28     dloc 1     ; deallocate parameter
29
30     halt
31 t:      dw      2 dup 0
32 @1:     dw      1
33 @spsave: dw      0
34     end      main

```


Are there pointers in Java?

Reference parameters in C++ hide the passing and dereferencing of an address. Java hides its pointers in a similar fashion.

On the next slide, both the C++ and Java programs dereference an address. In the C++ program, this process is explicit. In the Java program, this process is hidden.

FIGURE 8.18 a) C+ program with pointer p

```
1 class S {
2     public:
3     int x;
4     int y;
5 };
6 void main()
7 {
8     S *p;           // declare a pointer p
9     p = new S();    // create object and assign its address to p
10    p -> x = 3;      // (*p).x = 3;
11    p -> y = 4;      // (*p).y = 4;
12 }
```


b) Java program with reference p

```
1 class S {
2     int x;
3     int y;
4 }
5
6 class Pex {
7     public static void main(String arg[]) {
8         S p;           // create reference p
9         p = new S();    // assign p an instance of S
10        p.x = 3;        // access x via p
11        p.x = 4;        // access y via p
12    }
13 }
```


p and q are Java references. What happens when
`q = p;` is executed?

FIGURE 8.19

a)



b)



Pointers to functions

- Allowed in C++
- The C++ compiler translates the name of a function not followed by parenthesis to a pointer to that function (i.e., to its address). For example, **fp(3);** is a function call, but **fp** is a pointer to the fp function.

Declaring a pointer to a function

```
int (*p)(int x);
```

p is a pointer to a function that is passed an int and returns an int.

FIGURE 8.20

```
1 #include <iostream>
2 using namespace std;
3
4 int y;
5 int (*p)(int);
6 int fp(int x)
7 {
8     cout << x << endl;
9     return x;
10 }
11 void main()
12 {
13     y = fp(3);           // call fp
14     p = fp;
15     y = p(3);           // call fp via p pointer
16 }
```

$y = p(3);$

is difficult to translate, where p is a pointer to a function.

```
ld    @call
add   p           ; the required call inst now in ac
st    * + 3
ldc   3
push
dw    0           ; changes to a call
dloc  1
st    y
```

where

```
@call:  dw    call 0
```

Pointer arithmetic in C++

Arithmetic on pointers in C++ uses the size of the pointed to type as the unit size. For example, if `p` is a pointer to a struct consisting of 10 words, then

```
p = p + 1;
```

adds 10 (not 1) to `p`.

An array name is an address

The name of an array without the square brackets points to the 1st slot of the array. For example, given

```
int table[3];
```

Then `table` by itself is the address of `table[0]`. That is,

`table` (type is int pointer)

and

`&table[0]` (type is int pointer)

are equivalent.

Using array name as a pointer

The name of an array can be used as a pointer to that array (because it really is a pointer to the array). For example, given `int table[3];` then

`*(table+ 2) = 99;`

and

`table[2] = 99;`

are equivalent.

Using a pointer to an array as if it were the name of the array

A pointer to an array can be used as if it were the name of an array.

```
int *p;
```

```
int table[3];
```

```
p = table;          /* p now points to table
```

p and table have the same value and same type
so we can use them interchangeably (almost). */

```
p[2] = 99; // use p as if it were an array name.
```

```
int table[3];  
int *p;  
p = table;  
int x;
```

p is a pointer variable; table is a pointer constant.

So table cannot be assigned a new value.

```
p = table + 1;  // legal because p is a variable  
table = &x;     // illegal, table always points to table[0]
```

Creating a global array (static local array is similar)

```
int table[3];  
table[2] = 7;
```

is translated to

```
ldc 7  
st  table + 2
```

```
table: dw 3 dup 0
```

Creating a dynamic local array

```
void f() {  
    int table[3];  
    table[2] = 7;  
}
```

is translated to

```
alloc 3  
ldc 7  
str 2
```

Accessing dynamic local array

FIGURE 8.21




FIGURE 8.22

```
1 #include <iostream>
2 using namespace std;
3
4 int ga[3], x;           // ga is global array
5 void arrays()
6 {
7     static int sla[3];  // sla is a static local array
8     int dla[3];         // dla is a dyn local array
9
10    cout << "enter index\n";
11    cin >> x;
12
13    ga[2] = 99;
14    ga[x] = 99;
15
16    sla[2] = 99;
17    sla[x] = 99;
18
19    dla[2] = 99;
20    dla[x] = 99;
21
22 }
23 void main ()
24 {
25     arrays();
26 }
```


Must compute the address of
ga[x]

at run time because the value of
x is not known until run time.

FIGURE 8.23

```
1 @arrays$v:
2         aloc 3                ; int arrays ga[3];
3
4         ldc  @m0              ; cout << "enter index\n";
5         sout
6
7         din                   ; cin >> x;
8         st    x
9
10        ldc  99                ; ga[2] = 99;
11        st    ga + 2
12
13        ldc  99                ; ga[x] = 99;
14        push
15        ldc  ga
16        add  x
```

(continued)

FIGURE 8.23
(continued)

```
17          sti
18
19          ldc  99          ; sla[2] = 99;
20          st   @s0_sla + 2
21
22          ldc  99          ; sla[x] = 99;
23          push
24          ldc  @s0_sla
25          add  x
26          sti
27
28          ldc  99          ; dla[2] = 99;
29          str  2
30
31          ; dla[x] = 99;
32          ldc  99          ; get 99
33          push          ; prepare for sti
34          swap          ; get sp
35          st   @spsave    ; save it
36          swap          ; restore sp
37          ldc  1          ; get rel address of dla[0]
38          add  @spsave    ; get abs address of dla[0]
39          add  x          ; get abs address of dla[x]
40          sti          ; store 99 into dla[x]
41
42          dloc 3
43          ret
44 ;=====
45 main:      call @arrays$v ; arrays();
46
47          halt
48 ga: slg dw  3 dup 0      ; global array
49 @s0_slg: dw  3 dup 0      ; static local array
50 x:        dw  0
51 @m0:      dw  "enter index\n"
52 @spsave:   dw  0
53          end  main
```

Passing an array—really passing a pointer to 1st slot

```
int table[3];
```

```
...
```

```
    f(table);
```

Passing a pointer to the first slot of table.
Corresponding argument should be an int
pointer.

Empty square brackets in an array parameter are interpreted by the compiler as meaning pointer

```
void f(int t[ ]) {           // t is an int *  
    ...  
}
```


Arrays as arguments

FIGURE 8.24

```
1 int table[3];  
2 void tabfun(int t[]) // t is parameter  
3 {  
4     t[2] = 5;  
5 }  
6 void main()  
7 {  
8     tabfun(table); // array table is an argument  
9 }
```

t is a pointer to table[0]. It can be used as a pointer, or as if it were the name of the array it is pointing to.

FIGURE 8.25



Equivalent forms

FIGURE 8.26 a)

```
void tabfun(int *t)
{
    *(t+2) = 5;           // use t as an int pointer
}
```

b)

```
void tabfun(int *t)
{
    t[2] = 5;             // use t as an array name
}
```

c)

```
void tabfun(int t[])
{
    *(t + 2) = 5;         // use t as an int pointer
}
```


FIGURE 8.27

```
1 @tabfun$pi:
2                               ;t[2] = 5;
3         ldc    5                ; get 5
4         push                   ; prepare for sti
5         ldc    2                ; get index 2
6         addr   2                ; get address of t[2]
7         sti                      ; store 5 in t[2]
8
9         ret
10 ;=====
11
12 main:    ldc    table           ; call tabfun(table);
13         push
14         call   @tabfun$pi
15         dloc   1
16
17         halt
18 table:   dw     3 dup 0
19         end    main
```

Control statements

- while
- do-while
- for
- if
- if-else

Assembly code for while

```
while (x) {      // assume x is a global int
    cout << x << endl;
    x++;
}
```

where **x** is a global variable is

```
@L0:  ld x
      jz @L1      ; jump on false

      ld x        ; cout << x << endl;
      dout
      ldc '\n'
      aout

      ldc 1        ; x++
      add x
      st x


      ja @L0      ; jump always
@L1:
```

while

FIGURE 8.28 C++ form


```
while (expression)
    S;
```

Assembly form



do-while

```
do {  
    s;  
} while (expression);
```



for

```
for (s1; expression; s2)  
    s3;
```

Code for s1

Evaluate expression ←

Jump on false to

Code for s3

Code for s2

Jump always to



If-else

```
if (expression)
```

```
    s1;
```

```
else
```

```
    s2;
```


Evaluate expression

Jump on false to

Code for s1

Jump always to

Code for s2




if

```
if (expression)  
    s;
```


Evaluate expression

Jump on false to

Code for s



Better translation of while



This form has the same size as the form in Figure 8.28; however, each iteration of the loop now requires the execution of only one jump instruction (the jump on true) instead of two. Thus, this form runs faster. Rewriting the previous **while** loop in this form, we get

```
ja @L2          ; jump always

@L3: ld  x          ; cout << x << endl;
     dout
     ldc  '\n'
     aout

     ldc  1          ; x++
     add x
     st  x

@L2: ld  x
     jnz @L3          ; jump on true
```

Signed and unsigned comparisons

- Cannot be done easily on H1 because H1 does not have the S, V, and C flags.
- Subtracting two signed numbers and looking at the sign of the result yields the incorrect answer if overflow occurs.
- For now, we will assume overflow does not occur when performing comparisons.

Error in signed comparison

Compare 8000 and 0001 by subtracting and testing the sign of the result:

8000 (-32,768)

FFFF (-1)

7FFF

Positive result implies (incorrectly) that the top number is bigger than the bottom number.

Incorrect assembly code (wrong if overflow occurs)

FIGURE 8.29

C++	Assembly Language
if (x>=y) {	ld x
	sub y
	jn @LO
.	.
.	.
.	.
}	@LO:

Multi-word addition

Cannot be done easily on H1.

$z = x + y;$

where x , y , and z are double words is translated to

ld $x + 1$

add $y + 1$

st $z + 1$

ld x

add y

st z

; what if carry out?

Bit-level operations

- Cannot be done easily on H1.
- Only jzop and jn are available on H1 for bit-level operations. These instructions test the msb in the ac register.

A program that converts decimal to binary.

How to perform '&' operation on line 13?

FIGURE 8.30

```
1 #include <iostream>
2 using namespace std;
3
4 void main()
5 {
6     int x;
7     int count = 16;                // count = number bits in int
8     int mask = 0x8000;             // mask has only leftmost bit= 1
9     cout << "Enter a decimal number: ";
10    cin >> x;
11    cout << "Binary equivalent = ";
12    for (int i = 1; i <= count; i++) {
13        if (x & mask)                // bitwise AND; leftmost bit == 1?
14            cout << 1;
15        else
16            cout << 0;
17        x = x << 1;                  // left shift x one position
18    }
19    cout << endl;
20 }
```

A recursive function is a function
that calls itself.

What is the output of the program
on the next slide?

FIGURE 8.31

```
1 #include <iostream>
2 using namespace std;
3
4 void frec(int x)                // recursive function
5 {
6     if (x == 0)
7         cout << "B\n";
8     else {
9         cout << "D\n";
10        frec(x - 1);            // recursive call
11        cout << "U\n";
12    }
13 }
14 void main()
15 {
16     frec(2);
17 }
```

Output of program on preceding line is

D

D

B


U

U

Statements preceding the recursive call
are executed “on the way down”.

Statements following the recursive call
are executed “on the way up”.

FIGURE 8.32



When the compiler translates a recursive function to assembly language, it does not treat a recursive function call in any special way. For example, the recursive call within the **frec** function

```
frec(x - 1);
```

is translated in the usual way:


```
ldr 1      ; get value of x
sub @1     ; compute value of x - 1
push       ; create parameter on stack (the next x)
call @frec$i
dloc 1     ; remove the parameter from the stack
```


FIGURE 8.33

```
1  @frec$i: ldr 1 ; if (x == 0)
2           jnz @L0
3
4           ldc @m0 ; cout << "b\n";
5           sout
6
7           ja @L1
8
9  @L0:     ldc @m1 ; cout << "D\n";
10          sout
11
12          ldr 1 ; frec(x - 1);
13          sub @1
14          push
15          call @frec$i
16          dloc 1
17
18          ldc @m2 ; cout << "U\n";
19          sout
20
21  @L1:     ret
22  ;=====
23  main:    ldc 2 ; frec(2);
24          push
25          call @frec$i
26          dloc 1
27
28          halt
29  @m0:     dw "B\n"
30  @m1:     dw "D\n"
31  @m2:     dw "U\n"
32  @1:      dw 1
33          end main
```

ldr 1

at the beginning of freq
(line 1 of program on preceding
slide) always loads the
appropriate x.








FIGURE 8.34

Starting session. Enter h or ? for help.

---- [T7] F: ldc /8 000/ **b0**

Machine-level breakpoint set at 0

---- [T7] F: ldc /8 000/ n

No display mode

---- [T1] g

Machine-level breakpoint at 0

---- [T1] **d\$**

FFE: 0012 0002 ← first x

---- [T1] **g**

D

Machine-level breakpoint at 0

---- [T1] **d\$**

FFC: 000B 0001 0012 0002 ← first x

---- [T1] **g**

D



Machine-level breakpoint at 0000

---- [T1] **d\$**


FFA: 000B 0000 000B 0001 0012 0002 ← first x

---- [T1] **g**

B

U

U



Machine inst count = 23 (hex) = 35 (dec)

---- [T1] **q**